

## Tutorial Series

# Cartan Subalgebras, Compact Roots and the Satake Diagram for $su(2, 2)$

## Table of Contents

Synopsis	6. The Cartan Subalgebra II
1. The Matrix Algebra $su(2, 2)$	7. Classification
2. Dynkin and Satake Diagrams	Commands Illustrated
3. The Killing Form	References
4. The Cartan Subalgebra I	Release Notes
5. The Cartan Decomposition	Author

## Related Pages

Cartan Subalgebras

### Synopsis

In this worksheet we use the 15-dimensional real Lie algebra  $su(2, 2)$  to illustrate some important points regarding the general structure theory and classification of real semi-simple Lie algebras.

1. Recall that a real semi-simple Lie algebra  $\mathfrak{g}$  is called a compact Lie algebra if the Killing form is negative definite. The Lie algebra  $\mathfrak{g}$  is compact if and only if all the root vectors for any Cartan subalgebra are purely imaginary. However, if the root vectors are purely imaginary for some choice of Cartan subalgebra it is not necessarily true that the Lie algebra is compact.

2. A real semi-simple Lie algebra  $\mathfrak{g}$  is called a split Lie algebra if there exists a Cartan subalgebra such that the root vectors are all real. However, it is not true that if the root vectors are all real with respect one choice of Cartan subalgebra then they are real with respect to any other choice.

3. Points 1 and 2 reflect the fact that, for complex Lie algebras, all Cartan subalgebras are equivalent in the sense that they may be mapped into each other by a Lie algebra automorphism. This is not true for real semi-simple Lie algebras.

4. To properly analyze the structure of a given real Lie algebra  $\mathfrak{g}$ , one must first calculate a Cartan decomposition  $\mathfrak{g} = \mathfrak{t} + \mathfrak{p}$ . The Killing form of  $\mathfrak{g}$  is negative-definite on  $\mathfrak{t}$  and positive-definite on  $\mathfrak{p}$ . Next one must chose a Cartan subalgebra  $\mathfrak{h}$  which [i] is aligned with the Cartan decomposition in the sense that  $\mathfrak{h} = (\mathfrak{h} \cap \mathfrak{t}) + (\mathfrak{h} \cap \mathfrak{p})$  and [ii] such that the non-compact part  $\mathfrak{h} \cap \mathfrak{p}$  is of maximal dimension. Such Cartan subalgebras are said to be maximally non-compact. With such a Cartan subalgebra one obtains the minimum number of pure imaginary (or compact) roots, one can draw the Satake diagram and thereby obtain the correct real classification of the Lie algebra.

We illustrate these points with the real Lie algebra  $su(2, 2)$ .

## 1. The Matrix Algebra $su(2, 2)$

```
[> with(DifferentialGeometry) : with(LieAlgebras) :
```

The matrix algebra  $su(2, 2)$  is the 15-dimensional real Lie algebra of  $4 \times 4$  complex matrices  $M$  which are trace-free and skew-hermitian with respect to

$$I_{22} = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix},$$

that is,

$$M \cdot I_{22} + I_{22} \cdot M^\dagger = 0.$$

Here is the standard basis for  $su(2, 2)$ . (See also StandardRepresentation.)

```
[> M1 := Matrix([[I,0,0,0],[0,0,0,0],[0,0,0,0],[0,0,0,-I]]) :
```

```

> M2 := Matrix([[0,0,0,0],[0,I,0,0],[0,0,0,0],[0,0,0,-I]]):
> M3 := Matrix([[0,0,0,0],[0,0,0,0],[0,0,I,0],[0,0,0,-I]]):
> M4 := Matrix([[0,I,0,0],[I,0,0,0],[0,0,0,0],[0,0,0,0]]):
> M5 := Matrix([[0,0,I,0],[0,0,0,0],[-I,0,0,0],[0,0,0,0]]):
> M6 := Matrix([[0,0,0,I],[0,0,0,0],[0,0,0,0],[-I,0,0,0]]):
> M7 := Matrix([[0,0,0,0],[0,0,I,0],[0,-I,0,0],[0,0,0,0]]):
> M8 := Matrix([[0,0,0,0],[0,0,0,I],[0,0,0,0],[0,-I,0,0]]):
> M9 := Matrix([[0,0,0,0],[0,0,0,0],[0,0,0,I],[0,0,I,0]]):
> M10 := Matrix([[0,1,0,0],[-1,0,0,0],[0,0,0,0],[0,0,0,0]]):
> M11 := Matrix([[0,0,1,0],[0,0,0,0],[1,0,0,0],[0,0,0,0]]):
> M12 := Matrix([[0,0,0,1],[0,0,0,0],[0,0,0,0],[1,0,0,0]]):
> M13 := Matrix([[0,0,0,0],[0,0,1,0],[0,1,0,0],[0,0,0,0]]):
> M14 := Matrix([[0,0,0,0],[0,0,0,1],[0,0,0,0],[0,1,0,0]]):
> M15 := Matrix([[0,0,0,0],[0,0,0,0],[0,0,0,1],[0,0,-1,0]]):
> su22Matrices := [M1, M2, M3, M4, M5, M6, M7, M8, M9, M10, M11, M12, M13, M14, M15];

```

$$\text{su22Matrices} := \left[ \begin{array}{c} \begin{bmatrix} I & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -I \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & I & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -I \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & I & 0 \\ 0 & 0 & 0 & -I \end{bmatrix}, \begin{bmatrix} 0 & I & 0 & 0 \\ I & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & I & 0 \\ 0 & 0 & 0 & 0 \\ -I & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & I \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -I & 0 & 0 & 0 \end{bmatrix}, \\ \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & I & 0 \\ 0 & -I & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & I \\ 0 & 0 & 0 & 0 \\ 0 & -I & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & I \\ 0 & 0 & I & 0 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 & 0 \\ -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & -1 & 0 \end{bmatrix} \end{array} \right], \quad (2.1)$$

It is easy to check directly that each of these matrices belongs to  $su(2, 2)$ . Here is one way to do it. First use the command [DGzip](#) to create the general element of  $su(2, 2)$ .

```
> vars := [a1, a2, a3, a4, a5, a6, a7, a8, a9, a10, a11, a12, a13, a14, a15];
      vars := [a1, a2, a3, a4, a5, a6, a7, a8, a9, a10, a11, a12, a13, a14, a15]
```

(2.2)

```
> M := DGzip(vars, su22Matrices, "plus");
```

$$M := \begin{bmatrix} I a1 & I a4 + a10 & I a5 + a11 & I a6 + a12 \\ I a4 - a10 & I a2 & I a7 + a13 & I a8 + a14 \\ -I a5 + a11 & -I a7 + a13 & I a3 & I a9 + a15 \\ -I a6 + a12 & -I a8 + a14 & I a9 - a15 & -I a1 - I a2 - I a3 \end{bmatrix}$$

(2.3)

Now define the matrix  $I_{22}$ .

```
> I22 := LinearAlgebra:-DiagonalMatrix([1, 1, -1, -1]);
```

$$I_{22} := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$$

(2.4)

Check the defining property  $M \cdot I_{22} + I_{22} \cdot M^\dagger = 0$ . For this calculation we shall need to explicitly indicate to Maple that the variables  $[a_1, a_2, \dots, a_{15}]$  are real. Note that Hermitian conjugation is denoted with an asterisk in Maple.

```
> M.I22 + I22.M^* assuming seq(a::real, a = vars);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

(2.5)

Alternatively, we can use the [Query](#) command with the keyword argument "MatrixAlgebra".

```
> Query(su22Matrices, "su(2, 2)", version = 2, "MatrixAlgebra");
```

2.6

*true*

(2.6)

We shall work with the abstract Lie algebra defined by these matrices. The command [LieAlgebraData](#) calculates the structure equations for the matrices *su22Matrices*. Initialize this Lie algebra using [DGsetup](#).

```
> LD := LieAlgebraData(su22Matrices, alg);
```

```
LD := [e1, e2]=0, [e1, e3]=0, [e1, e4]=-e10, [e1, e5]=-e11, [e1, e6]=-2e12, [e1, e7]=0, [e1, e8]=-e14, [e1, e9] (2.7)
      ]=-e15, [e1, e10]=e4, [e1, e11]=e5, [e1, e12]=2e6, [e1, e13]=0, [e1, e14]=e8, [e1, e15]=e9, [e2, e3]=0, [e2, e4
      ]=e10, [e2, e5]=0, [e2, e6]=-e12, [e2, e7]=-e13, [e2, e8]=-2e14, [e2, e9]=-e15, [e2, e10]=-e4, [e2, e11
      ]=0, [e2, e12]=e6, [e2, e13]=e7, [e2, e14]=2e8, [e2, e15]=e9, [e3, e4]=0, [e3, e5]=e11, [e3, e6]=-e12, [e3, e7
      ]=e13, [e3, e8]=-e14, [e3, e9]=-2e15, [e3, e10]=0, [e3, e11]=-e5, [e3, e12]=e6, [e3, e13]=-e7, [e3, e14
      ]=e8, [e3, e15]=2e9, [e4, e5]=-e13, [e4, e6]=-e14, [e4, e7]=-e11, [e4, e8]=-e12, [e4, e9]=0, [e4, e10]=
      -2e1+2e2, [e4, e11]=e7, [e4, e12]=e8, [e4, e13]=e5, [e4, e14]=e6, [e4, e15]=0, [e5, e6]=e15, [e5, e7]=e10,
      [e5, e8]=0, [e5, e9]=-e12, [e5, e10]=e7, [e5, e11]=2e1-2e3, [e5, e12]=-e9, [e5, e13]=e4, [e5, e14]=0, [e5,
      e15]=e6, [e6, e7]=0, [e6, e8]=e10, [e6, e9]=-e11, [e6, e10]=e8, [e6, e11]=-e9, [e6, e12]=2e1, [e6, e13]=0,
      [e6, e14]=e4, [e6, e15]=-e5, [e7, e8]=e15, [e7, e9]=-e14, [e7, e10]=-e5, [e7, e11]=e4, [e7, e12]=0, [e7, e13
      ]=2e2-2e3, [e7, e14]=-e9, [e7, e15]=e8, [e8, e9]=-e13, [e8, e10]=-e6, [e8, e11]=0, [e8, e12]=e4, [e8, e13
      ]=-e9, [e8, e14]=2e2, [e8, e15]=-e7, [e9, e10]=0, [e9, e11]=-e6, [e9, e12]=-e5, [e9, e13]=-e8, [e9, e14
      ]=-e7, [e9, e15]=-2e3, [e10, e11]=-e13, [e10, e12]=-e14, [e10, e13]=e11, [e10, e14]=e12, [e10, e15]=0,
      [e11, e12]=e15, [e11, e13]=e10, [e11, e14]=0, [e11, e15]=e12, [e12, e13]=0, [e12, e14]=e10, [e12, e15]=-e11,
      [e13, e14]=e15, [e13, e15]=e14, [e14, e15]=-e13
```

```
> DGsetup(LD);
```

*Lie algebra: alg*

(2.8)

Here is the 2-dimensional multiplication table for our Lie algebra.


```
> MultiplicationTable("LieTable");
```

alg	$e_1$	$e_2$	$e_3$	$e_4$	$e_5$	$e_6$	$e_7$	$e_8$	$e_9$	$e_{10}$	$\epsilon$
$e_1$	0	0	0	$-e_{10}$	$-e_{11}$	$-2e_{12}$	0	$-e_{14}$	$-e_{15}$	$e_4$	
$e_2$	0	0	0	$e_{10}$	0	$-e_{12}$	$-e_{13}$	$-2e_{14}$	$-e_{15}$	$-e_4$	
$e_3$	0	0	0	0	$e_{11}$	$-e_{12}$	$e_{13}$	$-e_{14}$	$-2e_{15}$	0	$-$
$e_4$	$e_{10}$	$-e_{10}$	0	0	$-e_{13}$	$-e_{14}$	$-e_{11}$	$-e_{12}$	0	$-2e_1 + 2e_2$	
$e_5$	$e_{11}$	0	$-e_{11}$	$e_{13}$	0	$e_{15}$	$e_{10}$	0	$-e_{12}$	$e_7$	$2e_1 - 2$
$e_6$	$2e_{12}$	$e_{12}$	$e_{12}$	$e_{14}$	$-e_{15}$	0	0	$e_{10}$	$-e_{11}$	$e_8$	$-$
$e_7$	0	$e_{13}$	$-e_{13}$	$e_{11}$	$-e_{10}$	0	0	$e_{15}$	$-e_{14}$	$-e_5$	
$e_8$	$e_{14}$	$2e_{14}$	$e_{14}$	$e_{12}$	0	$-e_{10}$	$-e_{15}$	0	$-e_{13}$	$-e_6$	
$e_9$	$e_{15}$	$e_{15}$	$2e_{15}$	0	$e_{12}$	$e_{11}$	$e_{14}$	$e_{13}$	0	0	$-$
$e_{10}$	$-e_4$	$e_4$	0	$2e_1 - 2e_2$	$-e_7$	$-e_8$	$e_5$	$e_6$	0	0	$- \epsilon$
$e_{11}$	$-e_5$	0	$e_5$	$-e_7$	$-2e_1 + 2e_3$	$e_9$	$-e_4$	0	$e_6$	$e_{13}$	
$e_{12}$	$-2e_6$	$-e_6$	$-e_6$	$-e_8$	$e_9$	$-2e_1$	0	$-e_4$	$e_5$	$e_{14}$	$- \epsilon$
$e_{13}$	0	$-e_7$	$e_7$	$-e_5$	$-e_4$	0	$-2e_2 + 2e_3$	$e_9$	$e_8$	$-e_{11}$	$- \epsilon$
$e_{14}$	$-e_8$	$-2e_8$	$-e_8$	$-e_6$	0	$-e_4$	$e_9$	$-2e_2$	$e_7$	$-e_{12}$	
$e_{15}$	$-e_9$	$-e_9$	$-2e_9$	0	$-e_6$	$e_5$	$-e_8$	$e_7$	$2e_3$	0	$- \epsilon$

## 2. Dynkin and Satake diagrams of type A3



The Lie algebra  $su(2, 2)$  is of rank 3 and of type  $A$ . Since the rank is 3, there are 3 simple roots which we label by  $\{\alpha_1, \alpha_2, \alpha_3\}$ . Here is the  $A_3$  [Dynkin diagram](#)



```
alg > DynkinDiagram("A3");
```



▪ Points 1 ——— Curve 1

The real forms of the complex simple Lie algebras are completely characterized by their [Satake Diagrams](#). The 5 real forms of type  $A_3$  have the following Satake diagrams.

<p><math>sl(4)</math></p> <ul style="list-style-type: none"> <li>• no black dots <math>\Rightarrow</math> no pure imaginary simple roots</li> <li>• no red lines <math>\Rightarrow</math> no conjugate simple root pairs</li> </ul>	<p><math>su(4)</math></p> <ul style="list-style-type: none"> <li>• all black dots <math>\Rightarrow</math> 3 pure imaginary simple roots</li> <li>• no red lines <math>\Rightarrow</math> no conjugate simple root pairs</li> </ul>	<p><math>su(3, 1)</math></p> <ul style="list-style-type: none"> <li>• 1 black dot (<math>\alpha_2</math>), 1 pure imaginary root</li> <li>• 1 red line <math>\Rightarrow \{\alpha_1, \alpha_3\}</math> a conjugate simple root pair</li> </ul>
<pre>alg &gt; SatakeDiagram("sl(4)");</pre>	<pre>alg &gt; SatakeDiagram("su(4, 0)");</pre> 	<pre>alg &gt; SatakeDiagram("su(3, 1)");</pre> 



`[alg >`


*su*(2, 2)

- no black dots  $\Rightarrow$  no pure imaginary simple roots
- 1 red line  $\Rightarrow \{\alpha_1, \alpha_3\}$  a conjugate simple root pair

*su*\*(4)


- 2 black dots  $\Rightarrow$  2 pure imaginary simple roots
- no red lines  $\Rightarrow$  no conjugate simple root pairs

`[alg > SatakeDiagram("su(2, 2)");`



`[alg >`

`[alg > SatakeDiagram("su*(4)");`



`[alg >`



We see that, for example,  $su(2, 2)$  is uniquely characterized as the rank 3 Lie algebra of type A with no pure imaginary simple roots and 1 conjugate pair of complex roots.

The goal of this worksheet is to start from the abstract real Lie algebra (2.7), and construct its Satake diagram, in other words, to obtain the real classification of the Lie algebra (2.7).

### 3. The Killing Form

While it is not strictly necessary to do so, it is nevertheless quite helpful to begin our classification by calculating the [Killing form](#) of our Lie algebra and its signature.

```
alg > B := KillingForm(alg);
B := - 16 01 ⊗ 01 - 8 01 ⊗ 02 - 8 01 ⊗ 03 - 8 02 ⊗ 01 - 16 02 ⊗ 02 - 8 02 ⊗ 03 - 8 03 ⊗ 01 - 8 03 ⊗ 02 - 16 03 ⊗ 03
      - 16 04 ⊗ 04 + 16 05 ⊗ 05 + 16 06 ⊗ 06 + 16 07 ⊗ 07 + 16 08 ⊗ 08 - 16 09 ⊗ 09 - 16 010 ⊗ 010 + 16 011 ⊗ 011
      + 16 012 ⊗ 012 + 16 013 ⊗ 013 + 16 014 ⊗ 014 - 16 015 ⊗ 015
```

(4.1)

The command [QuadraticFormSignature](#) returns a vector space decomposition of the Lie algebra. The Killing form is positive-definite on the first summand, on the second summand the Killing form is negative-definite, and on the third summand it is totally degenerate.

```
alg > PosNegNull := Tensor:-QuadraticFormSignature(B);
PosNegNull := [[e5, e6, e7, e8, e11, e12, e13, e14], [e1, e1 - 2 e3, e1 - 3 e2 + e3, e4, e9, e10, e15], [ ]]
```

(4.2)

```
alg > map(nops, PosNegNull);
[8, 7, 0]
```

(4.3)

We conclude that for any Cartan decomposition of our algebra  $\mathfrak{g} = \mathfrak{t} + \mathfrak{p}$ , the dimension of the compact part is  $\dim(\mathfrak{t}) = 7$  and the dimension of the non-compact part is  $\dim(\mathfrak{p}) = 8$ . In particular, we see that our Lie algebra is not

compact -- we can immediately conclude that our algebra is not  $su(4)$ .

#### 4. The Cartan Subalgebra I

By definition, a Cartan subalgebra of a complex semi-simple Lie algebra is any abelian subalgebra which is self-normalizing. Any such subalgebra will do for the classification of the given Lie algebra as a complex Lie algebra.

The Maple command [CartanSubalgebra](#) calculates a basis for the Cartan subalgebra of our Lie algebra.

```
> CSA := CartanSubalgebra();
                                CSA := [e1, e2, e3]
(5.1)
```

Since the Cartan subalgebra is 3-dimensional, the rank of the Lie algebra is 3. We now find the [root space decomposition](#) of the Lie algebra to be

```
alg > RSD := RootSpaceDecomposition(CSA);
RSD := table([[1, 1, 2 I] = e9 + I e15, [0, -1, 1] = e7 - I e13, [-1, -1, -2 I] = e9 - I e15, [1, 2 I, 1] = e8 + I e14, [2 I, 1, 1]
= e6 + I e12, [-2 I, -1, -1] = e6 - I e12, [-1, -2 I, -1] = e8 - I e14, [-1, 0, 1] = e5 - I e11, [1, -1, 0] = e4 + I e10,
[0, 1, -1] = e7 + I e13, [-1, 1, 0] = e4 - I e10, [1, 0, -1] = e5 + I e11])
(5.2)
```

The [roots](#) and positive roots are

```
alg > LieAlgebraRoots(RSD);
[[ [1, 0, 2 I], [0, -1, 1], [-1, -1, -2 I], [1, 2 I, 1], [2 I, 1, 1], [-2 I, -1, -1], [-1, -2 I, -1], [-1, 0, 1], [1, -1, 0], [0, 1, -1], [-1, 1, 0], [1, 0, -1] ],
(5.3)
```

```
alg > PR := PositiveRoots(RSD);
PR := [[ [1, 0, 2 I], [0, -1, 1], [-1, -1, -2 I], [1, 2 I, 1], [2 I, 1, 1], [-2 I, -1, -1], [-1, -2 I, -1], [-1, 0, 1], [1, -1, 0], [0, 1, -1], [-1, 1, 0], [1, 0, -1] ],
(5.4)
```

This illustrates a main point in the synopsis, namely, that one cannot conclude that the Lie algebra is compact just because

the roots are all purely imaginary vectors. In order to proceed, we need the Cartan decomposition of the Lie algebra.

## 5. The Cartan Decomposition

A [Cartan decomposition](#) can be computed from a representation of the Lie algebra or from a root space decomposition.

```
[alg > T, P := CartanDecomposition(CSA, RSD, PR);  
      T, P := [e1, e2, e3, e4, e9, e10, e15], [e5, e6, e7, e8, e11, e12, e13, e14]
```

 (6.1)

We can check explicitly that this is a valid Cartan decomposition. The Killing form on  $T$  is negative-definite,

```
[alg > BT := Killing(T);  
      BT :=
```

$$\begin{bmatrix} -16 & -8 & -8 & 0 & 0 & 0 & 0 \\ -8 & -16 & -8 & 0 & 0 & 0 & 0 \\ -8 & -8 & -16 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -16 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -16 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -16 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -16 \end{bmatrix}$$

```
]
```

 (6.2)

```
[alg > LinearAlgebra:-IsDefinite(-BT);  
      true
```

 (6.3)

The Killing form on  $P$  is positive-definite,

```
[alg > BP := Killing(P);
```

$$BP := \begin{bmatrix} 16 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 16 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 16 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 16 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 16 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 16 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 16 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 16 \end{bmatrix} \quad (6.4)$$

and the decomposition is [symmetric](#).

```
[alg > Query(T, P, "SymmetricPair");
                                     true] (6.5)
```

We use this Cartan decomposition to find a maximally non-compact Cartan subalgebra.

## 6. The Cartan Subalgebra II

Here is the Cartan decomposition  $\mathfrak{g} = \mathfrak{t} \oplus \mathfrak{p}$  from the previous section:

```
[alg > T, P;
                                     [e1, e2, e3, e4, e9, e10, e15], [e5, e6, e7, e8, e11, e12, e13, e14]] (7.1)
```

We can use the Cartan decomposition to calculate another choice of Cartan subalgebra. This time we shall obtain a Cartan subalgebra  $\mathfrak{h}$  such that  $\mathfrak{h} = (\mathfrak{h} \cap \mathfrak{t}) + (\mathfrak{h} \cap \mathfrak{p})$ .

Let's find the Cartan subalgebra we want "by-hand". We first pick a vector (say  $e_5$ ) and calculate the maximal abelian subalgebra in  $\mathfrak{p}$  which contains  $e_5$ .

```
[alg > MaximalAbelianSubalgebra([e5], P);] (7.2)
```

$$[e_5, e_8] \quad (7.2)$$

Now look for the Cartan subalgebra which contains  $e_5, e_8$ .

```
alg > CartanSubalgebra(contains = [e5, e8]);
[e1 - e2 + e3, e5, e8] (7.3)
```

As a cautionary remark, one should also check that the [adjoint matrices](#) for  $e_5, e_8$  are semi-simple (diagonalizable).

We get the same result if we use the keyword argument `cartandecomposition`.

```
alg > newCSA := CartanSubalgebra(cartandecomposition = [T, P]);
newCSA := [e1 - e2 + e3, e5, e8] (7.4)
```

Here is the new root space decomposition with respect to the new Cartan subalgebra (which is maximal non-compact).

```
alg > newRSD := RootSpaceDecomposition(newCSA);
newRSD := table([ [0, 2, 0] = e1 - e3 + e11, [0, 0, -2] = e2 - e14, [0, 0, 2] = e2 + e14, [-2 I, 1, 1] = e4 + I e6 - I e7 - e9 - I e10 + e12 + e13 + I e15, [0, -2, 0] = e1 - e3 - e11, [2 I, 1, -1] = e4 + I e6 + I e7 + e9 + I e10 - e12 + e13 + I e15, [2 I, 1, 1] = e4 - I e6 + I e7 - e9 + I e10 + e12 + e13 - I e15, [-2 I, 1, -1] = e4 - I e6 - I e7 + e9 - I e10 - e12 + e13 - I e15, [2 I, -1, -1] = e4 + I e6 - I e7 - e9 + I e10 - e12 - e13 - I e15, [-2 I, -1, 1] = e4 + I e6 + I e7 + e9 - I e10 + e12 - e13 - I e15, [2 I, -1, 1] = e4 - I e6 - I e7 + e9 + I e10 + e12 - e13 + I e15, [-2 I, -1, -1] = e4 - I e6 + I e7 - e9 - I e10 - e12 - e13 + I e15]) (7.5)
```

We choose a set of positive roots which are closed under complex conjugation. Note that there are no pure imaginary roots.

```
alg > newPR := PositiveRoots(newRSD, <I, 3, 7>);
newPR := [ [ 0 ], [ 0 ], [ -2 I ], [ 2 I ], [ -2 I ], [ 2 I ],
           [ 2 ], [ 0 ], [ 1 ], [ 1 ], [ -1 ], [ -1 ],
           [ 0 ], [ 2 ], [ 1 ], [ 1 ], [ 1 ], [ 1 ] ] (7.6)
```

The simple roots are

```
alg > newSR := SimpleRoots(newPR);
```

$$newSR := \left[ \begin{bmatrix} 0 \\ 2 \\ 0 \end{bmatrix}, \begin{bmatrix} -2I \\ -1 \\ 1 \end{bmatrix}, \begin{bmatrix} 2I \\ -1 \\ 1 \end{bmatrix} \right]$$

(7.7)

We shall use these result to complete the real classification of the Lie algebra

## 7. Classification

We first determine the [Cartan matrix](#) for our Lie algebra and then transform it to [standard form](#).

```
alg > CM := CartanMatrix(newSR, newRSD);
```

$$CM := \begin{bmatrix} 2 & -1 & -1 \\ -1 & 2 & 0 \\ -1 & 0 & 2 \end{bmatrix}$$

(8.1)

```
alg > S1, S2, S3 := CartanMatrixToStandardForm(CM);
```

$$S1, S2, S3 := \left[ \begin{bmatrix} 2 & -1 & 0 \\ -1 & 2 & -1 \\ 0 & -1 & 2 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}, "A3" \right]$$

(8.2)

From this output, we see that, as a complex Lie algebra, our algebra is either  $A_3$  (or  $D_3$ ). To identify the simple roots in the list  $newSR$ , we need to re-order them as indicated by the permutation matrix  $S_2$  in (8.2).



```
alg > Delta := DGzip(convert(S2, listlist), newSR, "plus");
```

$$\Delta := \left[ \begin{bmatrix} -2I \\ -1 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 2 \\ 0 \end{bmatrix}, \begin{bmatrix} 2I \\ -1 \\ 1 \end{bmatrix} \right]$$

(8.3)

The [Dynkin diagram](#) for our Lie algebra is therefore

```
alg > DynkinDiagram("A", 3);
```



Finally, we have that none of the simple roots are pure imaginary so that for the [Satake diagram](#) none of the roots are colored. The root  $\alpha_1$  is the complex conjugate of  $\alpha_3$  so that these are [Satake associates](#) and are to be connected by a red dashed line.


```
alg > Delta[1], SatakeAssociate(Delta[1], Delta);
```

$$\begin{bmatrix} -2I \\ -1 \\ 1 \end{bmatrix}, \begin{bmatrix} 2I \\ -1 \\ 1 \end{bmatrix}$$

(8.4)

The only possibilities are  $su(2, 2)$  or equivalently  $so(4, 2)$ .


```
alg > SatakeDiagram("su(2, 2)");
```



• Points 1 — Curve 1 - - Curve 2

```
alg > SatakeDiagram("so(4, 2)");
```





• Points 1 — Curve 1 - - Curve 2

## Commands Illustrated

[DGsetup](#), [LieAlgebraData](#), [CartanMatrixToStandardForm](#), [CartanSubalgebra](#), [CartanDecomposition](#), [Centralizer](#), [DynkinDiagram](#), [KillingForm](#), [PositiveRoots](#), [QuadraticFormSignature](#), [Query](#), [SatakeAssociate](#), [SatakeDiagram](#), [SimpleRoots](#), [RootSpaceDecomposition](#)

## References

1. A. Cap and J. Slovak, *Parabolic Geometries I. Background and General Theory*, Mathematical Surveys and Monographs **144**, American Mathematical Society.
2. S. Helgason, *Differential Geometry, Lie Groups and Symmetric Spaces*, Pure and Applied Mathematics **80**, Academic

Press.

3. [http://en.wikipedia.org/wiki/Cartan\\_decomposition](http://en.wikipedia.org/wiki/Cartan_decomposition)

4. [http://en.wikipedia.org/wiki/Satake\\_diagram](http://en.wikipedia.org/wiki/Satake_diagram)

## ▼ Release Notes

- This worksheet was compiled with Maple 17 and DG release USU1, available by request from [ian.anderson@usu.edu](mailto:ian.anderson@usu.edu) .

## ▼ Author

- Ian Anderson, *Department of Mathematics and Statistics, Utah State University*

December 16, 2014